ブラック・ショールズのプログラム

Masato SHIMURA
m.shimura@jcom.home.ne.jp

2000/12/16

ブラックショールズ理論は, 金融工学の金字塔と賞賛されている。非常に難解なこの編微分方程式をいきなりプログラムしたり、数値解析ソフトに放り込んでも良好な結果を得ることは難しい。
本稿は、石村貞夫 · 石村園子著「金融 · 証券のためのブラックショウルズ微分方程式　１９９９　東京図書」により同方程式の解のスクリプトを作成した.

1　ブラックショウルズ方程式

債権, 株式, 通貨等の資産 $W_1, W_2, W_3 ... W_p$ を組み合わせポートフォリオができる.
時点ｔの資産の価格, 保有単位数、資産の収益率を次の表のようにする.

価格	$W_1(t)$	$W_2(t)$	...	$W_p(t)$
単位数	$n_1(t)$	$n_2(t)$	...	$n_p(t)$
収益率	$R_1(t)$	$R_2(t)$	...	$R_p(t)$

ここで、ポートフォリオの価値　$W(t)$ は

$$W(t) = n_1(t)W_1(t) + n_2(t)W_2(t) + ... + n_p(t)W_p(t)$$

ポートフォリオの収益率　$R(t)$ は

$$R(t) = \frac{n_1(t)W_1(t)}{W(t)}R_1(t) + \frac{n_2(t)W_2(t)}{W(t)}R_2(t) + ... + \frac{n_p(t)W_p(t)}{W(t)}R_p(t)$$

であらわされる。
σ 株価ボラティリティー
μ 期待収益率
S 株価

株価 S が伊藤過程

$$dS = \mu S \cdot dt + \sigma S \cdot dZ$$

に従っているとき, 伊藤のレンマから株価 S による派生証券の価格 $f(S,t)$ の微分　df　は

$$df = \{\frac{\partial f}{\partial S} \cdot \mu S + \frac{\partial f}{\partial t} + \frac{1}{2}\frac{\partial^2 f}{\partial S^2} \cdot \sigma^2 S^2\}dt + \frac{\partial f}{\partial S} \cdot \sigma S \cdot dZ$$

に従う。
株価 S の株式を $\frac{\partial f}{\partial S}$ 単位買って、価格　$S(S,t)$ の派生商品を一単位売るときのポートフォリオ価格は,

$$\frac{\partial f}{\partial S} \cdot S - 1 \cdot f(S,t)$$

Δt 時間でのポートフォリオ変化量は,

$$\frac{\partial f}{\partial S} \cdot \Delta S - 1 \cdot \Delta f$$

となる。この $\Delta S, \Delta f$ に伊藤のレンマを代入し、さらに, 非危険利子率 ｒ　を加えて整理していくとブラックショウルズ偏微分方程式ができる。

ブラックショールズの偏微分方程式は

$$rf(S,t) = \frac{\partial f}{\partial t} + \frac{1}{2}\frac{\partial^2 f}{\partial S^2}\sigma^2 S^2 + r\frac{\partial f}{\partial S}S$$

この式は, 株価 S の派生証券の価格を $f(S,t)$ としたとき, $f(S,t)$ が満たすべき偏微分方程式であり、これを解くことにより, 派生証券 (株価オプション）の価格評価公式が求まる。

2　ブラックショウルズ方程式の解

ブラック・ショールズ方程式は最後に次の式が解となる。

$$f(S,t) = S \cdot N(\frac{u}{\sigma\sqrt{x}} + \sigma\sqrt{x}) - Xe^{rx} \cdot N(\frac{u}{\sigma\sqrt{x}})$$

これは、つぎのようにあらわすことが出来る.

$$C = S \cdot N(d1) - Xe^{-r(T-t)} \cdot N(d2)$$

$$u = \log\frac{S}{X} + (r - \frac{\sigma^2}{2})(T - t)$$

$$d_1 = \frac{u}{\sigma\sqrt{x}} + \sigma\sqrt{x}$$

$$d_2 = d_1 - \sigma\sqrt{T - t}$$

3　ブラックショールズのプログラム

ブラック・ショールズの計算には, 次の５のパラメーターが必要である。ボラティィリティーは分散で, 日経新聞のオプション価格の欄から求める。

ヨーロピアン●コールオプションの価格を求める例

項目	記号	計算例
現在の株価	S	14500
権利行使価格	X	14000
オプションの期間	T	2ヶ月
ボラティィリティー	σ	38%
非危険利子率	r	6%

$T - t = \frac{2}{12} = 0.1667$

$u = \log \frac{S}{X} + (r - \frac{\sigma^2}{2})(T - t) = \log \frac{14500}{14000} + (0.06 - \frac{0.38^2}{2}) * 0.1667 = 0.0331$

$\frac{u}{\sigma\sqrt{x}} + \sigma\sqrt{x} = \frac{0.0331}{0.38 * \sqrt{0.1667}} + 0.38 * \sqrt{0.1667} = 0.3685$

$\frac{u}{\sigma\sqrt{x}} = \frac{0.0331}{0.38 * \sqrt{0.1667}} = 0.2133$

標準正規分布の値を求める。
[1]

ここでは、$N(\frac{u}{\sigma\sqrt{X}} + \sigma\sqrt{x}), N(\frac{u}{\sigma\sqrt{x}})$ の値は, 鈴木義一郎「J 言語による統計解析　1996　森北出版」に左からの正規分布の値を求めるスクリプトの例があるので, これによった.

$N(0.3685) = 0.6437$

$N(0.2133) = 0.5845$

$f(S, t) = 14500 * 0.6737 - 14000 * e^{-0.06*0.1667} * 0.5845 = 1232.0844$

3.1 Script

```
NB. ========================================
NB. Black Scholes Model
NB. ========================================

NB. ************************************************
NB. Normal distribution
NB. Original script is written by G. Suzuki math p 81
NB. modifyed by M.Shimura
NB. ***********************************************

 nd=:( % &( %: 2p1))&([: ^ [: - [: -: *:@])  NB. same as  ndens

 ndfs2=: 3 : 0
(-: h * (nd 0) + nd y. ) + h * +/ nd ( >: i.249) * h =. | y. % 250
```

[1]正規分布の数値表の値は本により, 左から求めるもの、右から, 中央からの 3 種類ある

```
)

ndf=: 0.5&+@(**ndfs2)


NB. ============================================
NB. Black Scholes Model
NB. M.Shimura  Nov. 2000
NB. ussage: bs data
NB. data is list ( 5 block)
NB. 現在の株価 権利行使価格 オプションの期間 (月）ボラティリティ
（%） 非危険利子率（%）
NB. ex. data=. 14500 14000  2 38 6
NB. ============================================
bs =: 3 : 0
'a b c d e'=. y.
t=. c % 12
u=. (^. a % b) +  t* (e1=.e % 100) - -:(bor=.d % 100) ^2
p2=. u % (bor * %: t)
p1=. p2 + bor * %: t
n1=.ndf p1
n2=. ndf p2
bs=. (a * n1 ) -  b *n2 *( ^ (-e1) * t)
bs
)
```

3.2 実行例

```
   ndf 0.3685
0.64375

   ndf 0.2133
0.584454
```

```
   bs 14500 14000 2 38 6
1233.08
```

A B-S Model by E.McDonnell

```
Jforum: Black-Scholes formula

Date: Mon, 14 Oct 2002 16:13:59 EDT
From: Eemcd@aol.com
Reply-To: forum@jsoftware.com
To: forum@jsoftware.com

The argument to the Black-Scholes formulae for Call and Put options has five
items:

S - the current price of the asset
X - the strike price (price when option is to be exercised)
T - the time in years
r - the risk-free interest rate
v - the volatility, typically the standard deviation of S for the last 3 or 6
months

Hu the posted a suite of four programs to this forum last June to give the
result of Black-Scholes calls and puts. For the most part they followed those
written in most other programming languages, but did make use of several of
the functions that are unique to J. Oleg Kobchenko suggested a change that
used an array approach to compute the final results. I suggested a change
that used an array approach to computing the derivatives needed. Arthur
Whitney gave a K solution that permitted the same program to solve for either
call or put simply by the sign of the volatility parameter v.

Here is a function that uses all these.

The Black Scholes formula is essentially the difference of products, -/M*D

The derivatives d1 and d2 are replaced by a 2-item list D:

D =: N((^.S%X)+T*r(+,-)-:*:v)%v*%:T
```

where N is the cumulative normal distribution function.

The money item M consists of S, the current asset price, and X*^-r*T, the present value of the strike price X, at rate r and time T:

```
M =: S,X*^-r*T
```

The result is the difference of their products:

```
-/M*D   NB. on my computer, slightly faster than M -/ .*D
```

The overall function can then be written as:

```
BS =: monad define
'S X T r v' =. y.
-/(S,X*^-r*T)*N((^.S%X)+T*r(+,-)-:*:v)%v*%:T
)
```

For example, let y =: 60 65 0.25 0.08, and append v or -v depending on whether a call or put is desired:

call:

```
   BS y, 0.3
2.13337
```

put:

```
   BS y,_0.3
_5.84628
```

The version of N that I use is due to Ewart Shaw in Vector 18.4. He uses J's Hypergeometric conjunction and the second expression in Abramowitz & Stegun

7.1.21 for the error function erf:

```
erf=:(*&(%:4p_1) % ^@:*:) * [: 1 H. 1.5 *: NB. Ewart Shaw Vector 18.4
cnd =:N=: [: -: 1: + [: erf %&(%:2)   NB. CDF of N(0,1)
```

Altogether, the J solution is quite compact: 42 tokens for BS and 25 for N and erf.

Eugene McDonnell

B Compare 2 Scripts

```
   y
60 65 0.25 0.08

BS y,0.3
2.13337
BS y,_0.3
_5.84628
   bs 60 65 3 30 8
2.13337
   bs 60 65 3 _30 8
_5.84628
```